# PrSAT: Second Examples

## Philosophy 148

February 7, 2008

### First, load in the `PrSAT` package

See my **`PrSAT` website** for instructions on downloading and installing **`PrSAT`** (assuming you have *Mathematica* installed).

```
<< PrSAT`
```

### Example #0

Proving a couple of equivalence theorems about independence. First, define a non-equivalence operator (for simpler input):

```
In[45]:= X_ ≢ Y_ := LogicalExpand[Not[(X ∧ Y) || (! X ∧ ! Y)]];

In[51]:= PrSAT[{
           (Pr[X ∧ Y] == Pr[X] Pr[Y]) ≢ (Pr[X ∧ ¬ Y] == Pr[X] Pr[¬ Y])
          }]

         PrSAT::srchfail : Search phase failed; attempting FindInstance

Out[51]= { }

In[52]:= PrSAT[
          {
           (Pr[X ∧ Y] == Pr[X] Pr[Y]) ≢ (Pr[¬ X ∧ ¬ Y] == Pr[¬ X] Pr[¬ Y])
          }]

         PrSAT::srchfail : Search phase failed; attempting FindInstance

Out[52]= { }
```

### Example #1

The pairwise independence *vs* mutual independence counterexample involving the labeled tickets.

```
MODEL1 = PrSAT[
  {
   Pr[X] == 1/2,
   Pr[Y] == 1/2,
   Pr[Z] == 1/2,
   Pr[X ∧ Y ∧ Z] == 0,
   Pr[X ∧ Y] == 1/4,
   Pr[X ∧ Z] == 1/4,
   Pr[Y ∧ Z] == 1/4
  }]
```

$\{\{X \to \{a_2, a_5, a_6, a_8\}, Y \to \{a_3, a_5, a_7, a_8\}, Z \to \{a_4, a_6, a_7, a_8\}, \Omega \to \{a_1, a_2, a_3, a_4, a_5, a_6, a_7, a_8\}\},$
$\{a_1 \to \frac{1}{4}, a_2 \to 0, a_3 \to 0, a_4 \to 0, a_5 \to \frac{1}{4}, a_6 \to \frac{1}{4}, a_7 \to \frac{1}{4}, a_8 \to 0\}\}$

```
TruthTable[MODEL1]
```

| X | Y | Z | var | Pr |
|---|---|---|---|---|
| T | T | T | $a_8$ | 0 |
| T | T | F | $a_5$ | $\frac{1}{4}$ |
| T | F | T | $a_6$ | $\frac{1}{4}$ |
| T | F | F | $a_2$ | 0 |
| F | T | T | $a_7$ | $\frac{1}{4}$ |
| F | T | F | $a_3$ | 0 |
| F | F | T | $a_4$ | 0 |
| F | F | F | $a_1$ | $\frac{1}{4}$ |

We can use **EvaluateProbability** to check that this model is a counterexample to the claim in question:

```
EvaluateProbability[{
  Pr[X ∧ Y] == Pr[X] Pr[Y],
  Pr[X ∧ Z] == Pr[X] Pr[Z],
  Pr[Y ∧ Z] == Pr[Y] Pr[Z],
  Pr[X ∧ Y ∧ Z] == Pr[X] Pr[Y] Pr[Z]},
 MODEL1]
```

{True, True, True, False}

We can use **PrSAT** to find a *regular* countermodel to this claim, as follows:

```
MODEL2 = PrSAT[{
   Pr[X ∧ Y] == Pr[X] Pr[Y],
   Pr[X ∧ Z] == Pr[X] Pr[Z],
   Pr[Y ∧ Z] == Pr[Y] Pr[Z],
   Pr[X ∧ Y ∧ Z] ≠ Pr[X] Pr[Y] Pr[Z]}, Probabilities → Regular]
```

$$\{\{X \to \{a_2, a_5, a_6, a_8\}, Y \to \{a_3, a_5, a_7, a_8\}, Z \to \{a_4, a_6, a_7, a_8\}, \Omega \to \{a_1, a_2, a_3, a_4, a_5, a_6, a_7, a_8\}\},$$
$$\{a_1 \to \frac{84\,418 - 39\sqrt{4\,676\,097}}{56\,277}, a_2 \to \frac{-42\,296 + 39\sqrt{4\,676\,097}}{168\,831}, a_3 \to \frac{-42\,296 + 39\sqrt{4\,676\,097}}{168\,831},$$
$$a_4 \to \frac{-42\,296 + 39\sqrt{4\,676\,097}}{168\,831}, a_5 \to \frac{1}{999}, a_6 \to \frac{1}{999}, a_7 \to \frac{1}{999}, a_8 \to \frac{42}{169}\}\}$$

```
TruthTable[MODEL2]
```

| X | Y | Z | var | Pr |
|---|---|---|---|---|
| T | T | T | $a_8$ | $\frac{42}{169}$ |
| T | T | F | $a_5$ | $\frac{1}{999}$ |
| T | F | T | $a_6$ | $\frac{1}{999}$ |
| T | F | F | $a_2$ | $\frac{-42\,296+39\sqrt{4\,676\,097}}{168\,831}$ |
| F | T | T | $a_7$ | $\frac{1}{999}$ |
| F | T | F | $a_3$ | $\frac{-42\,296+39\sqrt{4\,676\,097}}{168\,831}$ |
| F | F | T | $a_4$ | $\frac{-42\,296+39\sqrt{4\,676\,097}}{168\,831}$ |
| F | F | F | $a_1$ | $\frac{84\,418-39\sqrt{4\,676\,097}}{56\,277}$ |

## ■ Example #2

A counterexample to transitivity of independence.  We can use **PrSAT** to automatically find such a counterexample:

```
MODEL3 = PrSAT[
  {
   Pr[X ∧ Y] == Pr[X] Pr[Y],
   Pr[Y ∧ Z] == Pr[Y] Pr[Z],
   Pr[X ∧ Z] ≠ Pr[X] Pr[Z]
  }, Probabilities → Regular]
```

$$\{\{X \to \{a_2, a_5, a_6, a_8\}, Y \to \{a_3, a_5, a_7, a_8\}, Z \to \{a_4, a_6, a_7, a_8\}, \Omega \to \{a_1, a_2, a_3, a_4, a_5, a_6, a_7, a_8\}\},$$
$$\{a_1 \to \frac{1}{999}, a_2 \to \frac{427}{285\,714}, a_3 \to \frac{71}{143}, a_4 \to \frac{427}{285\,714}, a_5 \to \frac{1}{999}, a_6 \to \frac{1}{999}, a_7 \to \frac{1}{999}, a_8 \to \frac{71}{143}\}\}$$

```
TruthTable[MODEL3]
```

| X | Y | Z | var | Pr |
|---|---|---|---|---|
| T | T | T | $a_8$ | $\frac{71}{143}$ |
| T | T | F | $a_5$ | $\frac{1}{999}$ |
| T | F | T | $a_6$ | $\frac{1}{999}$ |
| T | F | F | $a_2$ | $\frac{427}{285\,714}$ |
| F | T | T | $a_7$ | $\frac{1}{999}$ |
| F | T | F | $a_3$ | $\frac{71}{143}$ |
| F | F | T | $a_4$ | $\frac{427}{285\,714}$ |
| F | F | F | $a_1$ | $\frac{1}{999}$ |

We can use **EvaluateProbability** to check that this model is a counterexample to the claim in question:

```
EvaluateProbability[{
  Pr[X ∧ Y] == Pr[X] Pr[Y],
  Pr[Y ∧ Z] == Pr[Y] Pr[Z],
  Pr[X ∧ Z] == Pr[X] Pr[Z]},
 MODEL3]
```

```
{True, True, False}
```

## ■ Example #3

An example of *Simpson's Paradox*:

```
MODEL4 = PrSAT[
  {
   Pr[p ⋀ q | r] == Pr[p | r] Pr[q | r],
   Pr[p ⋀ q | ¬ r] == Pr[p | ¬ r] Pr[q | ¬ r],
   Pr[p ⋀ q] ≠ Pr[p] Pr[q]
  }, Probabilities → Regular]
```

$$\{\{p \to \{a_2, a_5, a_6, a_8\}, q \to \{a_3, a_5, a_7, a_8\}, r \to \{a_4, a_6, a_7, a_8\}, \Omega \to \{a_1, a_2, a_3, a_4, a_5, a_6, a_7, a_8\}\},$$
$$\{a_1 \to \frac{12}{91}, a_2 \to \frac{5}{42}, a_3 \to \frac{8}{65}, a_4 \to \frac{49\,373}{46\,838\,610}, a_5 \to \frac{1}{9}, a_6 \to \frac{49\,373}{2\,629\,536}, a_7 \to \frac{1}{38}, a_8 \to \frac{15}{32}\}\}$$

```
TruthTable[MODEL4]
```

| p | q | r | var | Pr |
|---|---|---|---|---|
| T | T | T | $a_8$ | $\frac{15}{32}$ |
| T | T | F | $a_5$ | $\frac{1}{9}$ |
| T | F | T | $a_6$ | $\frac{49\,373}{2\,629\,536}$ |
| T | F | F | $a_2$ | $\frac{5}{42}$ |
| F | T | T | $a_7$ | $\frac{1}{38}$ |
| F | T | F | $a_3$ | $\frac{8}{65}$ |
| F | F | T | $a_4$ | $\frac{49\,373}{46\,838\,610}$ |
| F | F | F | $a_1$ | $\frac{12}{91}$ |

We can use **EvaluateProbability** to check that this model is a counterexample to the claim in question:

```
EvaluateProbability[{
  Pr[p ⋀ q | r] == Pr[p | r] Pr[q | r],
  Pr[p ⋀ q | ¬ r] == Pr[p | ¬ r] Pr[q | ¬ r],
  Pr[p ⋀ q] == Pr[p] Pr[q]},
 MODEL4]
```

```
{True, True, False}
```

## ▪ Example #4

Hacking's "Odd Question #5" — *Base Rate Fallacy* example:

```
MODEL5 = PrSAT[
  {
   Pr[𝔼 | ℍ] == 0.8,
   Pr[𝔼 | ¬ ℍ] == 0.1,
   Pr[ℍ] == 0.01
  }]
```

```
{{𝔼 → {a2, a4}, ℍ → {a3, a4}, Ω → {a1, a2, a3, a4}}, {a1 → 0.891, a2 → 0.099, a3 → 0.002, a4 → 0.008}}
```

```
TruthTable[MODEL5]
```

| $\mathbb{E}$ | $\mathbb{H}$ | var | Pr |
|---|---|---|---|
| T | T | $a_4$ | 0.008 |
| T | F | $a_2$ | 0.099 |
| F | T | $a_3$ | 0.002 |
| F | F | $a_1$ | 0.891 |

We can use **EvaluateProbability** to calculate the "posterior" Pr(H | E):

```
EvaluateProbability[Pr[ℍ | 𝔼], MODEL5]
```

```
0.0747664
```